Mathematical Optimization in Finance: Closing the gap

Franz Nelißen

GAMS Software GmbH

FNelissen@gams.com

GOR 2003, Heidelberg

Overview

- Portfolio Optimization Models
- Modeling Approaches
- Example
- Stochastic Programming
- Summary

GAMS

Mathematical Optimization in Finance

- Very active research field with significant contributions and important practical applications
- Some of the reasons:
  - Continual stream of challenging problems with obvious impact of uncertainty
  - High availability of data
  - Validation potential – benchmarking
  - Very competitive and liquid markets → Many instruments and strategies
  - $$

Portfolio Optimization Models

- Mean-Variance Analysis
- Portfolio Models for Fixed Income
- Scenario Optimization
- Indexation Models
- Stochastic Programming

Modeling Approaches

- General programming languages
  →C++, Delphi, FORTRAN, Java, VBA, …
- Spreadsheets
- **Algebraic Modeling Languages (AML)**
- Mixture of different approaches

Algebraic Modeling Languages (AML)

- **Declarative** approach
  - Implementation of the optimization problem is close to its mathematical formulation:
    - Variables, constraints with arbitrary names
    - Sets, indices, algebraic expressions, powerful sparse index and data handling
    - Efficient but simple syntax
  - Model formulation contains no hints how to process the model → AML translates this representation into another form suitable for the optimization algorithm
- Also **procedural elements**: Loops, procedures, macros, …

Design Features

- Separation of model and data
  → Problem structure is data independent
- Separation of model and solution methods
  →Multiple model types and solvers
  - LP, MIP, NLP, QP, QPMIP, MINLP,…
  - Links to various commercial codes and to research codes (open solver interface)
- Models are scalable and platform independent

Example

- Simple Mean Variance Model

$$Min \sum_{i=1}^{I} \sum_{j=1}^{J} x_i Q_{i,j} x_j$$

$$s.t. \quad \sum_{i=1}^{I} \mu_i x_i \geq M$$

$$\sum_{i=1}^{I} x_i = 1, \qquad x_i \geq 0$$

- GAMS Formulation:

```
vbal..    v =e= sum((i,j), x(i)*q(i,j)*x(j));
mbal..          sum(i, mu(i)*x(i)) =g= M;
budget..        sum(i, x(i))       =e= 1;
 x.lo(i)                             = 0;
```

Example - Formulation

```
$eolcom #
Set      i    analyzed investments;
Alias (i,j) ;
Parameter  mu(i)   expected return,
           q(i,j)  covariance matrix,
           mup     target expected return for the portfolio;
Variables  v       variance of portfolio,
           M       mean return of the portfolio,
           x(i)    fraction of the portfolio that consists of i;
Equations vbal     variance definition,
          mbal     mean balancing constraint,
          budget   budget constraint ;
  vbal..      v =e= sum((i,j), x(i)*q(i,j)*x(j));
  mbal..            sum(i, mu(i)*x(i)) =g= M;
  budget..          sum(i, x(i))       =e= 1;
```

Example - Formulation cont'd.

```
$include data.inc # get data from external file
* some bounds
M.lo    =  mup; # target return of the portfolio
x.lo(i) = 0;    # no short selling

Model var1 / vbal, mbal, budget / ;
Solve var1 minimizing v using nlp ;
display v.l, M.l, mup, x.l;
```

Example - Data

```
Set i / cn,fr,gr,jp,sw,uk,us    /;
Parameter  mu(i)/
 cn  0.1287,  fr   0.1096,  gr  0.0501,   jp     0.1524,
 sw  0.0763,  uk   0.1854,  us  0.0620                  /;
Table  q(i,j)
            cn       fr       gr       jp       sw       uk       us
    cn     42.18
    fr     20.18    70.89
    gr     10.88    21.58    25.51
    jp      5.30    15.41     9.60    22.33
    sw     12.32    23.24    22.63    10.32    30.01
    uk     23.84    23.80    13.22    10.46    16.36    42.23
    us     17.41    12.62     4.70     1.00     7.20     9.90    16.42 ;
Scalar  mup   / 0.115 /;
q(i,j)$(ord(j) gt ord(i)) = q(j,i) ;
```

Example - Solution

```
VARIABLE  v.L  =   10.487  variance of the portfolio
VARIABLE  m.L  =    0.115  mean return of the portfolio
PARAMETER mup  =    0.115  target expected return for the portfolio

VARIABLE x.L  fraction of the portfolio that consists of security i
   gr 0.014,    jp 0.465,    uk 0.090,    us 0.430
```

Example - Extensions

- Short Sales
- Efficient Frontiers
- Trading Restrictions (“Zero or Range“ – Constraints)
  → MINLP Problem

Example - Trading Restrictions

```
Table bdata(i,pd) portfolio data and trading restrictions
*                    - increase -          - decrease -
            old      umin    umax          lmin    lmax
     cn     0.2      0.03    0.11          0.02    0.30
     fr     0.2      0.04    0.10          0.02    0.15
     gr       0      0.04    0.07          0.04    0.10
     jp       0      0.03    0.11          0.04    0.10
     sw     0.2      0.03    0.20          0.04    0.10
     uk     0.2      0.03    0.10          0.04    0.15
     us     0.2      0.03    0.10          0.04    0.30
  ;
```

Trading Restrictions – Formulation

```
Variables xi(i)      fraction of portfolio increase
          xd(i)      fraction of portfolio decrease
          y(i)       binary switch for increasing current holdings of i
          z(i)       binary switch for decreasing current holdings of i;
Binary Variables  y, z; positive variables xi, xd;
Equations  maxinc(i)  bound of maximum lot increase of fraction of i,
           mininc(i)  bound of minimum lot increase of fraction of i,
           maxdec(i)  bound of maximum lot decrease of fraction of i,
           mindec(i)  bound of minimum lot decrease of fraction of i,
           binsum(i)  restrict use of binary variables,
           xdef(i)    final portfolio definition;
xdef(i)..     x(i)   =e=  bdata(i,'old') - xd(i) + xi(i);
maxinc(i)..   xi(i)  =l=  bdata(i,'umax')* y(i);
mininc(i)..   xi(i)  =g=  bdata(i,'umin')* y(i);
maxdec(i)..   xd(i)  =l=  bdata(i,'lmax')* z(i);
mindec(i)..   xd(i)  =g=  bdata(i,'lmin')* z(i);
binsum(i)..   y(i) + z(i) =l= 1;
Model var2  /all/; Solve var2 minimizing v using minlp ;
```

Trading Restrictions – Solution

```
VARIABLE  v.L = 15.418 variance of the portfolio
VARIABLE  m.L =  0.115 mean return of the portfolio
PARAMETER mup =  0.115 target expected return for the portfolio

----    147 PARAMETER report  summary report
```

				Increase		Decrease	
	old	**new**	**delta**	min	max	min	max
cn	**0.200**	**0.056**	**-0.144**	0.030	0.110	0.020	0.200
fr	**0.200**	**0.050**	**-0.150**	0.040	0.100	0.020	**0.150**
gr		**0.070**	**0.070**	0.040	**0.070**	0.040	0.100
jp		**0.110**	**0.110**	0.030	**0.110**	0.040	0.100
sw	**0.200**	**0.122**	**-0.078**	0.030	0.200	0.040	0.100
uk	**0.200**	**0.292**	**0.092**	0.030	0.100	0.040	0.150
us	**0.200**	**0.300**	**0.100**	0.030	**0.100**	0.040	0.200

Recent Developments

- Support of codes, which take advantage of special problem structures:
  - →QP / QPMIP via Interior Point Methods
  - →Conic Programming
- Support of various global optimization codes

Stochastic Programming (SP)

- Models so far are static: Decision is made, then not further modified
- Stochastic Programming models allow sequence of decisions
- Elements:
  - **Scenarios**: Complete set of possible discrete realizations of the uncertain parameters with probabilities
  - **Stages**: Decisions points. First stage decisions now, second stage decision (depending of the outcome of the first stage decision) after a certain period and so on

→**Recourse**: Describes how decision variables can adept to the different out comes of the random parameters at each stage

A simple Scenario Tree



A more complex Scenario Tree



Challenges

- Deterministic equivalent: Includes all scenarios and stages of SP

→Size of model explodes

- Challenges (among others):
  - Generation difficult
  - Solution may not be possible
  - Interpretation and validation of results

→Less applications of SP than one may expect

Challenges *cont'd.*

But: Number of uncertain parameters is small

- Efficient representation of the uncertain data within the AML?
- Some scenarios only differ slightly → Can we reduce the number of scenarios?
- Problems are structured → How can we take advantage of the different specialized solution techniques (Decomposition) for SP's

SP & AML

- Representation of uncertain data structures → New language elements necessary:
  - Special expressions and conventions for stages and scenario trees
  - Random distributions for some problem data
- Support of scenario reduction techniques can dramatically reduce the size of deterministic equivalent
- Automatic translation of problem description into input format for various SP-solvers (OSL SE, DECIS, …)
  →**But:** Different approaches, not yet clear which standard will be adopted

Summary

- Finance is a success story for OR applications
- Large classes of problems can be solved without major problems
- AML provide a powerful and flexible framework for these classes of models
- Stochastic programming still challenging but a lot of promising developments